

Sierpiński triangle construction using chaos game algorithm

ZX81 BASIC 10-Liner implementation

Marco Varesio

<https://retrobits.altervista.org/>

<https://retrobits.itch.io/>

<https://somebitsofme.altervista.org/>



Sierpinski10Liner is a simple [Sinclair ZX81 BASIC](#) language implementation of the [Sierpiński triangle](#) fractal construction using the [chaos game algorithm](#).

The program is made up of 10 (actually 9, the last is just a REMark statement) BASIC lines with one statement per line and is the improved version of [a similar program I made some years ago](#). It should run on any un-expanded (1KB RAM) or expanded ZX81 computer or clone.

***Bonus:** for a real retro-experience (with a modern web browser), [here](#) You can see the TELETEXT version of the BASIC listing and program output!*

Program description

According to the related [Wikipedia article](#), the chaos game algorithm for drawing the Sierpiński triangle can be summarized as follows:

1. Take three points in a plane to form a triangle, you need not draw it.
2. Randomly select any point inside the triangle and consider that your current position.
3. Randomly select any one of the three vertex points.
4. Move half the distance from your current position to the selected vertex.
5. Plot the current position.

6. Repeat from step 3.

The 3 triangle vertices are: (X=0, Y=0), (X=30, Y=40) and (X=0, Y=60). These values have been chosen as a result of the trade-off between output image quality and memory usage, since the program is meant to run with just 1KB of RAM.

As a further implementation simplification, the starting point is not randomly selected as described in step 2 but is fixed: (X=30, Y=40).

The following paragraphs explain program code in detail, line by line.

Current point initialization:

```
1 LET X=30
2 LET Y=40
```

Randomly select a number between 0 and 2 and store it to K variable. This value will be used to select one of the 3 vertex points:

```
3 LET K=INT (RND*3)
```

Select one of the 3 vertex points based on the value of K and store its coordinates in variables A and B:

```
4 LET A=30*K
5 LET B=40*(K=1)
```

In fact:

- $K=0 \rightarrow$ Select vertex (A=0, B=0): $A=30*K=30*0=0$; $B=40*(K=1)=40*0=0$.
- $K=1 \rightarrow$ Select vertex (A=30, B=40): $A=30*K=30*1=30$; $B=40*(K=1)=40*1=40$.
- $K=2 \rightarrow$ Select vertex (A=60, B=0): $A=30*K=30*2=60$; $B=40*(K=1)=40*0=0$.

Move half the distance from your current position to the selected vertex:

```
6 LET X=INT ((X+A)/2)
7 LET Y=INT ((Y+B)/2)
```

Plot the current position:

```
8 PLOT X, Y
```


Repeat from step 3:

```
9 GOTO 3
```


Emulator instructions

The following instructions explain how to load and run the program using [EightyOne](#) emulator on Windows systems.

- 1) [Download](#), unzip and start the emulator by executing EightyOne.exe:



- 2) Select "Hardware..." from "Options" menu and make sure that ZX81 computer is selected in the window that pops up:



- 3) Select "Open Tape..." from "File" menu to locate and load **Sierpinski10Liner.p** file
- 4) In the next seconds, if the emulator is configured in order to automatically load tape images, you should see something happening on the screen and finally a white screen with only the "0/0" message on the bottom:

0/0

If so, you can skip to step 6); if nothing happens, you must manually start tape image loading as described in step 5).

- 5) Tap the "J" key. You should see "LOAD" followed by a black cursor on the bottom of the screen. Press and hold "SHIFT" key and tap "P" key twice. You should see 2 double quotation marks after the "LOAD" message:

LOAD ""

Release the "SHIFT" key and press "ENTER". After some seconds, you should see something happening on the screen and finally a white screen with the "0/0" message on the bottom, as described in step 4).

- 6) You can now either run the program or see its BASIC listing.

- a. **To RUN the program:** tap the “R” key. You should see “RUN” followed by a black cursor on the bottom of the screen:


RUN █

Press “ENTER” key to start the program. The (emulated) ZX81 will start painting the Sierpiński triangle, dot by dot.

- b. **To LIST the program:** tap the “K” key. You should see “LIST” followed by a black cursor on the bottom of the screen:

LIST █

Press “ENTER” key to show program code:



The screenshot shows a window titled "EightyOne" with a menu bar containing "File", "View", "Control", "Options", "Tools", and "Help". The main area displays the following program code:

```
1 LET X=30
2 LET Y=40
3 LET K=INT (RND*3)
4 LET A=30*K
5 LET B=40*(K=1)
6 LET X=INT ((X+A)/2)
7 LET Y=INT ((Y+B)/2)
8 PLOT X,Y
9 GOTO 3
10 REM MARCO VARESIO 2020
```

At the bottom left of the window, it shows "0/0". At the bottom of the window, there are two status indicators: "ZX81" and "50fps".

On *nix systems, you can use the [sz81 emulator](#).